CS61A Notes – Week 9: Mutation

Revenge of the Box-and-pointers

Consider this:

```
> (define x (list 1 2 3))
> (define y (list 1 2 3))
> (define z x)
```

1. What does (eq? x y) return? **#f**

2. What does (eq? z x) return? **#t**

3. How many lists are created in total? **2**

Because you guys don't do the lab...

Take a look at this procedure from *SICP*, exercise 3.14:

```
(define (mystery x)
   (define (loop x y)
      (if (null? x)
           y
           (let ((temp (cdr x)))
              (set-cdr! x y)
              (loop temp x))))
   (loop x '()))

(define v (list 'a 'b 'c 'd))
(define w (mystery v))
```

To help clarify, `loop` uses the "temporary" variable `temp` to hold the old value of the `cdr` of `x`, since the `set-cdr!` on the next line destroys the `cdr`.

QUESTION

Draw box-and-pointer diagrams that show the structures `v` and `w` after evaluating those expressions. What does Scheme print for the values of `v` and `w`?

```
w => (d c b a)
v => (a)
```

Teenage Mutant Ninja... erm, Schemurtle (you try to do better)

QUESTIONS

1.Personally – and don't let this leave the room – I think `set-car!` and `set-cdr!` are useless; we can just implement them using `set!`. Check out my two proposals for `set-car!`. Do they work, or do they work? Prove me wrong:

a.
```
(define (set-car! thing val)
   (set! (car thing) val))
```

`Doesn't work – set! is a special form! It cannot evaluate what (car thing) is.`

b. **(define (set-car! thing val)**
```
   (let ((thing-car (car thing)))
      (set! thing-car val)))
```

Doesn't work. thing-car is a new symbol bound to the value of (car thing), and the set! statement simply sets the value of thing-car to val, without touching the original thing at all.

2.I'd like to write a procedure that, given a deep list, destructively changes all the atoms into the symbol `colleen`:
```
> (define ls '(1 2 (3 (4) 5)))
> (glorify! ls) => return value unimportant
> ls => (colleen colleen (colleen (colleen) colleen))
```
Here's my proposal:
```
(define (glorify! L)
   (cond ((atom? L)
           (set! L 'colleen))
          (else (glorify! (car L))
                (glorify! (cdr L)))))
```

Does this work? Why not? Write a version that works.

No. Remember, to manipulate elements of a list, you need to use set-car! or set-cdr!. set! sets ls to 'scheme when ls is an atom, but once we return, the new value for ls is lost. Here's a way to do this:
```
      (define (glorify! ls)
         (cond ((null? ls) '())
               ((atom? ls) 'colleen)
               (else (set-car! ls (glorify! (car ls)))
                     (set-cdr! ls (glorify! (cdr ls)))
                     ls)))
```

We need to return ls because the set-car! and set-cdr! expressions expect glorify! to return something – namely, the transformed sublist.

3.We'd like to rid ourselves of odd numbers in our list:
`(define my-lst '(1 2 3 4 5))`
a.Implement `(no-odd! ls)` that takes in a list of numbers and returns the list without the odds, using mutation: `(no-odd! my-lst) => '(2 4)`

`(no-odd! my-lst) => '(2 4)`

```
(define (no-odd! ls)
   (cond ((null? ls) '())
         ((odd? (car ls)) (no-odd! (cdr ls)))
         (else (set-cdr! ls (no-odd! (cdr ls)))
         ls)))
```

b.Implement `(no-odd! ls)` again. This time, it still takes in a list of numbers, but can return anything. But after the call, the original list should be mutated so that it contains no odd numbers. Or,
```
(no-odd! my-lst) => return value unimportant
my-lst => '(2 4)
```
Hint: Try to consider if this is possible before you start!

This, I fear, is not possible. Note that we need to skip the first element (the 1), so we'd like to do something like (set! my-lst ...) to have it point to the solution instead of the cons pair containing 1 as the car. However, inside the procedure no-odd!, we have no way of doing that; we can set! ls to all our heart's content, but we have no way of altering the value of my-lst. Make sure you understand why; this point is crucial.

There is hope – with use of sentinels. If we always represent lists like so:

```
(define (new-list first . rest)
   (cons #f (cons first rest)))
```

where (new-list 1 2 3 4 5) => (#f 1 2 3 4 5), then we can do this:

```
(define (no-odd! ls)
   (cond ((null? ls) '())
         ((null? (cdr ls)) ls)
         ((odd? (cadr ls)) (set-cdr! ls (no-odd! (cddr ls))) ls)
         (else (set-cdr! ls (no-odd! (cdr ls))) ls)))
```

Carefully consider how the new representation of lists allowed us to do that. What we wanted to be able to do was to have my-lst point to something else (other than the pair containing 1). But we had no way to do that. So instead, we have my-lst point to some useless thing – like a pair containing #f – and, with that, we'll be able to set! the first element of the list to something else. If that's clear as mud, draw a few box-and-pointer diagrams!

We used the word "sentinel" to refer to #f. A "sentinel" is a meaningless value at the start of a data structure that allows us to do what we did above. It also makes the code look prettier, as in the remove! procedure presented in lecture.

It would also be nice to have a procedure which, given a list and an item, inserts that item at the end of the list by making only one new cons cell. The return value is unimportant, as long as the element is inserted. In other words,

```
> (define ls '(1 2 3 4))
> (insert! ls 5) => return value unimportant
> ls => (1 2 3 4 5)
```

4.Implement our old friend's ruder cousin, `(reverse! ls)`. It reverses a list using mutation. (This is a standard programming job interview question.)

```
;; here's a way using state
(define (reverse! ls)
   (define (helper! prev rest)
      (cond ((null? rest) prev)
            (else (let ((rest-of-rest (cdr rest)))
               (set-cdr! rest prev)
               (helper! rest rest-of-rest)))))
   (helper! '() ls))

;; this way is similar to the way we approached the old reverse
;; problem; reverse! the rest of the list, and then attach the
;; first element to the last
(define (reverse! ls)
   (define (last ls)
      (cond ((null? ls) '())
            ((null? (cdr ls)) ls)
            (else (last (cdr ls)))))
   (cond ((null? ls) '())
         ((null? (cdr ls)) ls)
         (else (let ((rest-reversed (reverse! (cdr ls))))
                  (set-cdr! ls '())
                  (set-cdr! (last rest-reversed) ls)
                  rest-reversed))))
```

5.Implement `(interleave! ls1 ls2)` that takes in two lists and interleaves them without allocating new `cons` pairs.

```
(define (interleave! ls1 ls2)
   (cond ((null? ls1) ls2)
         ((null? ls2) ls1)
         (else (set-cdr! ls1 (interleave! ls2 (cdr ls1)))
                 ls1)))
```

EXTRA PRACTICE

6.It would also be nice to have a procedure which, given a list and an item, inserts that item at the end of the list by making only one new `cons` cell. The return value is unimportant, as long as the element is inserted. In other words,

```
> (define ls `(1 2 3 4))
> (insert! ls 5) => return value unimportant
> ls => (1 2 3 4 5)
```

Does the following procedure work? If not, can you write one that does?

```
(define (insert! L val)
   (if (null? L)
       (set! L (list val))
       (insert! (cdr L) val)))
```

Nope; this should be automatic by now: set! does not change elements or structure of a list! As for glorify!, you might try returning partial answers like this:

```
(define (insert! ls val)
   (cond ((null? ls) (list val))
         (else (set-cdr! ls (insert! (cdr ls) val))
               ls)))

(define ls '(1 2 3 4))
(insert! ls 5) => (1 2 3 4 5)
ls => (1 2 3 4 5)
```

This almost works. But what if ls is null?

```
(define ls '())
(insert! ls 3) => (3)
ls => '()
```

Why doesn't this work? Think carefully.

7.Write a procedure, `remove-first!` which, given a list, removes the first element of the list destructively. You may assume that the list contains at least two elements. So,

```
> (define ls `(1 2 3 4))
> (remove-first! ls) => return value unimportant
> ls => (2 3 4)
```

And what if there's only one element?

```
(define (remove-first! L)
   (cond ((null? (cdr (cdr L)))
          (set-car! L (cadr L))
          (set-cdr! L '()))
         (else (set-car! L (cadr L))
               (remove-first! (cdr L)))))
```

Note that having at least two elements in the list allows us to do (cdr (cdr ls)) in the base case. If there's only one element, this wouldn't work, and in fact there's no general procedure that can remove the first element given any list. Consider:

```
(define ls '(1))
(remove-first! ls)
ls => '()
```

This means we'll need to change what ls points to (from a cons cell to a null list), which we know we can't do from within remove-first!

8.Implement `(deep-map! proc deep-ls)` that maps a procedure over every element of a deep list, without allocating any new cons pairs. So,

```
   (deep-map! square `(1 2 (3 (4 5) (6 (7 8)) 9))) =>
      `(1 4 (9 (16 25) (36 (49 64)) 81))
```

```
(define (deep-map! proc ls)
   (cond ((null? ls) '())
         ((not (pair? ls)) (proc ls))
         (else (set-car! (deep-map! proc (car ls)))
               (set-cdr! (deep-map! proc (cdr ls))))))
```